

## Physics 350 Lab 11 (Nyquist Frequency) Sample Solutions

For this lab, we are including a sample solution here consisting of two IDL procedure files. If you place these two files in your `~/idl` directory and then execute the command (within IDL)

```
lab11_sols,/ps
```

and you will create a postscript file called `idl.ps` containing the images attached to this file.

The `lab11_sols.pro` file contains the following:

```
PRO lab11_sols, PS_PLOTS = PsPlots
;
;
; The stuff below sets the output to postscript if desired.
  IF keyword_set(PsPlots) THEN BEGIN
    OrigDevice = !d.name
    set_plot, 'ps'
  ENDIF
;
; This next bit sets 4 plots per window. I did this to simplify
; putting the solutions together, but you were not expected to.
!p.multi = [0, 2, 2]

;
; Problem 1
;
N = 50L
DeltaX = 2*!pi/N
x = DeltaX*findgen(N+1)
kf = 2*!pi/(N*DeltaX)
knyq = (N/2)*kf

xfine = 2*!pi/10000*findgen(10001)

;
; part a
;
; The first plot I'll write out all the commands below.

plot, x, cos(kf*x), xrange=[0, 2*!pi], /XSTYLE ; This one looks good.

; for the rest I made a procedure so I didn't have to
; type the same stuff over and over again.

lab11_sin_cos_plot, kf, 10, x ;This one looks OK too
lab11_sin_cos_plot, kf, 20, x ;Definitely not good.
lab11_sin_cos_plot, kf, 15, x ;Marginal, not a perfect cosine.

; My answer for part (a) is 15, but a bit higher or lower is also
; reasonable.
```

## Physics 350 Lab 11 (Nyquist Frequency) Sample Solutions

```
;
; part b
;

lab11_sin_cos_plot, kf, 1, x, /SINE
lab11_sin_cos_plot, kf, 10, x, /SINE
lab11_sin_cos_plot, kf, 15, x, /SINE
lab11_sin_cos_plot, kf, 20, x, /SINE

; My answer here is also 15, so both are the same.

;
; Problem 2
; part a
;
; set to three plots per page now.
;
!p.multi = [0, 1, 3] ; you didn't have to do this in your solutions.
;
; First we do this for N/2+1
lab11_sin_cos_plot, kf, N/2+1, x
lab11_sin_cos_plot, kf, N/2+1, xfine, TITLE_TEXT="exact"
plot, xfine, cos((N/2+1)*kf*xfine), XRANGE=[0, 2*!pi], /XSTYLE, $
      XTITLE="x", YTITLE='cos', TITLE='n=26, both'
oplot, x, cos((N/2+1)*kf*x), psym=4
;
; Wow, doesn't look like a cosine at all. But the sampled function
; makes sense when overplotted.

; Now we try N/2+10
lab11_sin_cos_plot, kf, N/2+10, x
lab11_sin_cos_plot, kf, N/2+10, xfine, TITLE_TEXT="exact"
plot, xfine, cos((N/2+10)*kf*xfine), XRANGE=[0, 2*!pi], /XSTYLE, $
      XTITLE="x", YTITLE='cos', TITLE='n=35, both'
oplot, x, cos((N/2+10)*kf*x), psym=4
;
; Interesting. This one looks like a cosine, but with completely the
; wrong wavelength.

;
; and now N/2+20, which is 45....almost up to N
;
lab11_sin_cos_plot, kf, N/2+20, x
lab11_sin_cos_plot, kf, N/2+20, xfine, TITLE_TEXT="exact"
plot, xfine, cos((N/2+20)*kf*xfine), XRANGE=[0, 2*!pi], /XSTYLE, $
      XTITLE="x", YTITLE='cos', TITLE='n=26, both'
oplot, x, cos((N/2+20)*kf*x), psym=4
;
```

## Physics 350 Lab 11 (Nyquist Frequency) Sample Solutions

```
; Awesome! The sampled function is a really nice looking cosine.
; But the wavelength is really, really wrong.
; The overplotted graph shows we are sampling correctly, but with just
; the right timing to mimic a lower frequency wave. What we get out has
; as much to do with how we are sampling as it does with the
; function we are trying to sample.

;
; part b
;
; We are so far off because we are sampling too infrequently to catch
; the rapid changes in the function we are trying to sample.
;

;
; part c
;
;
; OK, now two plots per page....
;
!p.multi = [0, 1, 2]

lab11_sin_cos_plot, kf, N+1, xfine, TITLE_TEXT="exact"
lab11_sin_cos_plot, kf, N+1, x

;
; part d
;
;
;
; The sampled graph looks exactly like a wave with the fundamental
; frequency! Note sure it is exactly the same? I'll overplot the
; fundamental as triangles:
oplot, x, cos(kf*x), PSYM=4

;stuff below closes the postsrript file
IF keyword_set(PsPlots) THEN BEGIN
    device, /close
    set_plot, OrigDevice
ENDIF
;
; Set plots per window back to 1:
;
!p.multi = 0

END
```

## Physics 350 Lab 11 (Nyquist Frequency) Sample Solutions

The *lab11\_sin\_cos\_plot.pro* file contains the following:


```
PRO lab11_sin_cos_plot, kf, n, x, SINE_PLOT = Sine, TITLE_TEXT =
TitleWords

  IF n_elements(TitleWords) EQ 0 THEN TitleWords = ""
  IF keyword_set(Sine) THEN y = sin(n*kf*x) ELSE y = cos(n*kf*x)
  IF keyword_set(Sine) THEN SineOrCosine = 'Sine' $
  ELSE SineOrCosine = 'cosine'


  plot, x, y, XRANGE=[0, 2*!pi], /XSTYLE, $
    XTITLE='x', YTITLE=SineOrCosine, $
    TITLE="n="+strtrim(n, 2)+" "+TitleWords
  RETURN
END
```

# Physics 350 Lab 11 (Nyquist Frequency) Sample Solutions

Plots related to Problem 1a (from within *idl.ps* file created by **lab11\_sols,/ps**):




Plots related to Problem 1b (from within *idl.ps* file created by **lab11\_sols,/ps**):




# Physics 350 Lab 11 (Nyquist Frequency) Sample Solutions

Plots related to Problem 2a (from within *idl.ps* file created by **lab11\_sols**, /ps):



Physics 350 Lab 11 (Nyquist Frequency) Sample Solutions



Plots related to Problem 2c (from within *idl.ps* file created by `lab11_sols,/ps`):

